
1: $PROBLEM    PARE/VALDECOXIB IN CHILDREN
2: 
3: $INPUT      ID TIME AMT RATE DV CMT MDV TYPE AGE WT SEX PROP MRPH PRCT FENT ATCR ONDN DEXA ABIO
4: 
5: $DATA       PAREBLQDATA.CSV IGNORE=#
6: 
7: $SUBROUTINE ADVAN5 TRANS1
8: 
9: $PRIOR      NWPRI NTHETA=14 NETA=4 NEPS=2 NTHP=2 NETP=0 NPEXP=1

10: 
11: $MODEL      COMP=PAREN ;1
12:             COMP=META ;2
13:             COMP=PAR2 ;3
14:             COMP=PAR3 ;4
15: 
16: $PK
17: ;TYPICAL(POPULATION) ESTIMATES FOR PARECOXIB
18: TVCL1=THETA(3)*((WT/70)**0.75)*(1+THETA(10)*(AGE-6.9))
19: TVV1 =THETA(4)*WT/70
20: TVQ3=THETA(5)*((WT/70)**0.75)
21: TVV3=THETA(6)*WT/70
22: TVQ4=THETA(7)*((WT/70)**0.75)
23: TVV4=THETA(8)*WT/70
24: 
25: ;ERROR(%) ESTIMATE FOR PARECOXIB
26: SIG=THETA(9)
27: 
28: ;TYPICAL(POPULATION) ESTIMATES FOR VALDECOXIB PARAMETERS INCORPORATING MATURATION
29: TVHILL=THETA(1)
30: HILL=TVHILL
31: TVMA50=THETA(2)
32: MA50=TVMA50
33: FCL=((AGE+.74615)**HILL)/(((AGE+.74615)**HILL)+MA50**HILL)
34: TVCL2=THETA(11)*FCL*((WT/70)**0.75)
35: 
36: TVHIL2=THETA(13)
37: HILL2=TVHIL2
38: TVMA52=THETA(14)
39: MA502=TVMA52
40: FCL2=((AGE+.74615)**HILL2)/(((AGE+.74615)**HILL2)+MA502**HILL2)
41: TVV2=THETA(12)*FCL2*WT/70
42: 
43: ;INDIVIDUAL ESTIMATES FOR PARECOXIB RELATED PARAMETERS
44: CL1=TVCL1*EXP(ETA(3))
45: V1=TVV1 
46: Q3=TVQ3      
47: V3=TVV3*EXP(ETA(4))
48: Q4=TVQ4   
49: V4=TVV4
50: ;INDIVIDUAL ESTIMATES FOR VALDECOXIB RELATED PARAMETERS INCORPORTATING MOLECULAR RATIO
51: CL2=TVCL2*(314.36/370.42)*EXP(ETA(1))
52: V2=TVV2*(314.36/370.42)*EXP(ETA(2))
53: 
54: ;CUSTOM RATES BETWEEN COMAPARTMENTS
55: K12=CL1/V1           
56: K31=Q3/V3          
57: K13=Q3/V1          
58: K41=Q4/V4          
59: K14=Q4/V1       
60: K20=CL2/V2     
61: 
62: ;SCALE TERMS FOR COMAPARTMENTS      
63: S1=V1                 
64: S2=V2              
65: S3=V3 
66: S4=V4          
67: 
68: $ERROR 
69: ; ERROR MODEL FOR PARECOXIB INCORPORATING BLQ DATA
70: IF(CMT.EQ.1) THEN 
71: LOQ=LOG(10)
72: IPRED=LOG(F+0.001)
73: IRES=DV-IPRED
74: IWRES=IRES/1
75: DUM=(LOQ-IPRED)/SIG
76: CUMD=PHI(DUM)
77: ENDIF
78: IF(TYPE.EQ.1.AND.CMT.EQ.1) THEN
79: F_FLAG=0

80: Y=LOG(F)+SIG*ERR(1)
81: ENDIF
82: IF(TYPE.EQ.2.AND.CMT.EQ.1) THEN
83: F_FLAG=1
84: Y=CUMD
85: ENDIF
86: 
87: ; ERROR MODEL FOR VALDECOXIB
88: IF(CMT.EQ.2) THEN 
89: IPRE =LOG(F+0.001)
90: IRES=DV-IPRE
91: IWRE=IRES/1
92: Y=LOG(F)+ERR(2) 
93: ENDIF
94: 
95: ;INITIAL ESTIMATES OF THETA
96: $THETA  
97:  (0,2.4,100)  ; HILL-CL(VALDECOXIB)/F*
98:  (-1,1.0,4)   ; MA50-CL(VALDECOXIB)/F*
99:  (0,21)       ; CL(PARECOXIB)

100:  (0,5.4)      ; VC(PARECOXIB)
101:  (0,7.0)      ; Q1(PARECOXIB)
102:  (0,83)       ; VP1(PARECOXIB)
103:  (0,2.7)      ; Q2(PARECOXIB)
104:  (0,4.9)      ; VP2(PARECOXIB)
105:  (0,0.31)     ; ERROR(%) ESTIMATE FOR PARECOXIB
106:  (0,0.021,.3) ; MATURATION SLOPE (%) CL(PARECOXIB)
107:  (0,11)       ; CL(VALDECOXIB)/F*
108:  (0,87)       ; V(VALDECOXIB)/F*
109:  (0,2.6,100)  ; HILL-V(VALDECOXIB)/F*
110:  (-1,0.71,4)  ; MA50-V(VALDECOXIB)/F*
111:  
112: ; PRIOR DATA FOR THETA(1)&(2)
113: $THETA  
114:  3 FIX       ; HILL-CL(VALDECOXIB)/F*
115:  1.4115 FIX  ; MA50-CL(VALDECOXIB)/F*
116:  
117: ;INITIAL ESTIMATES FOR ETAS AND COVARIANCES
118: $OMEGA  BLOCK(4) 
119:  0.1  ; CL2
120:  0.05 0.1  ; V2
121:  0 0.05 0.1  ; CL1
122:  0 0 0.05 0.1  ; V3
123:  
124: ;PRIOR DATA FOR THETA(1)&(2) 
125: $OMEGA  BLOCK(2) FIX 
126:  0.011207     ; HILL-CL(VALDECOXIB)/F*
127:  0  0.338724  ; MA50-CL(VALDECOXIB)/F*
128:  
129: ;INITIAL ESTIMATES FOR ERR TERMS
130: $SIGMA 
131:  1  FIX  ; PARECOXIB
132:  0.0360  ; VALDECOXIB
133: 
134: $ESTIMATION MAX=9999 SIG=2 METHOD=COND INTER LAPLACIAN NUMERICAL POSTHOC PRINT=1 MSFO=MSF1
135:             
136: $COVARIANCE PRINT=E MATRIX=R
137: 




